# $Id: PKGBUILD 83124 2013-01-27 16:09:09Z pierre $
# Maintainer: Kevin Piche <kevin@archlinux.org>
# Contributor: Ewoud Nuyts <ewoud.nuyts@gmail.com>

pkgname=bmp-wma
pkgver=0.1.1
pkgrel=4
pkgdesc="WMA plugin for Beep Media Player"
url="http://bmp-plugins.berlios.de/novelian/project.php?p=bmp-wma"
license=('GPL')
arch=('x86_64' 'i686')
depends=('bmp')
groups=('bmp-plugins' 'bmp-io-plugins')
options=('!libtool' '!emptydirs')
source=("http://download.berlios.de/bmp-plugins/${pkgname}-${pkgver}.tar.gz"
        'bmp-wma-gcc4.patch')
md5sums=('8dcf3fcdb10e8e2e386f70745812412b'
         '730eba8f41e989dd8b56c18bcc826cc4')

build() {
  cd "${srcdir}/${pkgname}-${pkgver}"
  patch -Np0 -i "${srcdir}/bmp-wma-gcc4.patch"
  if [ "${CARCH}" = "x86_64" ]; then
    export CFLAGS="${CFLAGS} -fPIC"
  fi
  ./configure --prefix=/usr --disable-static
  echo '#define HAVE_LRINTF 1' >> config.h
  make
}

package() {
  cd "${srcdir}/${pkgname}-${pkgver}"
  make DESTDIR="${pkgdir}" install
}
